
Basic Model Equations of Team Dynamics – Base Run

Level  Acceptance_Div_Ideas
init
Acceptance_Div_Ideas = 20

flow
Acceptance_Div_Ideas = -dt*Loss_Of_AcceptDiv


+dt*Ch_in_Accept

doc
Acceptance_Div_Ideas = The degree in which members tolorate diversity in ideas    

            and personalities, etc. This is a stock of openness. At the moment this is   

            principally an aggregated level or stock in the model.

Level   Burnout

init
Burnout = 0

flow
Burnout = -dt*DissOf_Brnout


+dt*BrnoutGeneration

doc
Burnout = This fatigue process is modeled as a first order material delay.

Level
Care Respect

init
Care_Respect = 10

flow
Care_Respect = +dt*ChIn_CareForTeam_Members

doc
Care_Respect = How strongly members care for each other.  This is a stock that represents a more emotional side of team members.

Level
Clarity_Of_Vision

init
Clarity_Of_Vision = 10

flow
Clarity_Of_Vision = +dt*IncInVision


-dt*DecrIn_Vision

doc
Clarity_Of_Vision = Deals with how clear the vision of the team is to the team members. This model does not deal with the overall vison of the company as a whole and the potential gap between the companies vision and that of the team. 

Level
Commitment

init
Commitment = 20

flow
Commitment = -dt*Loss__of_Commit


+dt*Gain_Of_Commit

doc
Commitment = A key stock in the model reflecting the readiness and motivation of potential team members to accept the concept of a self-directed team.

Level
Empowerment

init
Empowerment = 10

flow
Empowerment = +dt*In_In_Empowr


-dt*LossOf_Empowr

doc
Empowerment = The feeling of self-efficacy and ability to be free to take on new challenges.

Level
External_Resources

init
External_Resources = .25

flow
External_Resources = -dt*DecIn_ExtResrs_1


+dt*IncInExt_Resrs

doc
External_Resources = Resources, such as time, money, and equipment from sources outside of the company.  

Level
Indicated_Goals

init
Indicated_Goals = 60

flow
Indicated_Goals = +dt*IncrIn_TLevAspir


-dt*Erosion_Of_Goal

doc
Indicated_Goals = Goals that are indicated by the general level of competition in the business environment. Note that this is without considerating the resources needed to meet the team's goals.

Level
InfAbout_CBaseYCompet
init
InfAbout_CBaseYCompet = 10

flow
InfAbout_CBaseYCompet = +dt*IncIn__Inform


-dt*LossOf_Inform

doc
InfAbout_CBaseYCompet = This state variable represents External Information of client base and what the competitors are doing.  I have initially set it quite low because before forming self-directed teams, information was held "close to the vest".  Currently it is being modeled as a first order material delay.

Level
Internal_Resources

init
Internal_Resources = 25

flow
Internal_Resources = +dt*ChangeIn_Res

doc
Internal_Resources = Resources, such as time, money, and equipment from sources internal to the company.  

Level   Knowledge

init
Knowledge = 20

flow
Knowledge = +dt*Learning_about_New_tasks


-dt*Loss_Of__Knowledge

doc
Knowledge = Knowledge about self-directed teams. This is non-experiential accumulated knowledge for the most part.

Level
Level_of_Tm_Function

init
Level_of_Tm_Func = 10

flow
Level_of_Tm_Func = -dt*Dec_InTm_Func


+dt*Inc_In_Tm_Funct

doc
Level_of_Tm_Func = Level of Team Functioning.  Overall indicator of how well the team is working with each other, with the rest of the organization, and with the external custormer/client base. Note:  Increases in the Level_of_Tm_Func feeds upon itself, and thus has a self-loop.  This is consistent with Mihaly Csikszentmihalyi's "flow" concept.

Level
Loyalty

init
Loyalty = 20

flow
Loyalty = -dt*OutRt_Loyal


+dt*InR_Loyalt

doc
Loyalty = The degree of loyalty towards the team, it members, as well as loyalty to the company.  At the moment, this is an aggregated stock.

Level
Memb_Skills

init
Memb_Skills = 10

flow
Memb_Skills = -dt*DecInTLrdrshpSkills


+dt*IncrInTlrdsshpSkills

doc
Memb_Skills = An aggregated stock of skills that are related to the tasks done by the team over time.

Level
Member_Ambiguity
init
Member_Ambiguity = 90

flow
Member_Ambiguity = -dt*Loss_of_Mem_Amb

doc
Member_Ambiguity = his version of the model assumes that preteam performance generates much ambiguity, which can decrease over time as the potential team members obtain knowledge of self-directed teams.

Level
Member_Communication
init
Member_Communication = 20

flow
Member_Communication = -dt*Loss_of_Commun


+dt*Gain_In_MembComm

doc
Member_Communication = The degree in which members are able to communicate with each other, the rest of the organization, and to the outside custormer/clients/supplier stakeholders.

Level
Mutual_Accountability
init
Mutual_Accountability = 20

flow
Mutual_Accountability = -dt*Decr_In_Acc


+dt*Inc_Acc

doc
Mutual_Accountability = The degree that members hold each other accountable for their actions and performance  behavior.

Level
Performance
init
Performance = 40

flow
Performance = +dt*InIc_Perf


-dt*LossOf_Perf


+dt*PreTeam_PerfChRate

doc
Performance = This is the generic performance level or attainment that would vary from team to team and company to company.

Level
Team_Spirit
init
Team_Spirit = 20

flow
Team_Spirit = -dt*LossOf_TSpirt


+dt*IncIn_TSpirit

doc
Team_Spirit = An intense identification with the team  in which work is a vocation and people on the team have a large sense of connectedness with other team members.  This is congruent with Russ Moxley's use of the term "spirit."

Level
Team_Zeal
init
Team_Zeal = 20

flow
Team_Zeal = +dt*GainZl


-dt*Loss_of_Zl

doc
Team_Zeal = The degree of intense happiness and dedication  in performing tasks as a team member.

Level
TeamSize
init
TeamSize = Desired_Team_Size

flow
TeamSize = +dt*Gen_Hire_Rate


-dt*Quit_Fire_Rate

doc
TeamSize = The size of the team

Level
Trust_In_Team_Ldr
init
Trust_In_Team_Ldr = 20

flow
Trust_In_Team_Ldr = -dt*DecreaseIn_Team_Ldr


+dt*Inc_In_Team_Ldr

doc
Trust_In_Team_Ldr = The amount of trust the team members have in the team leader. Very important especially in the formation phase.

Level
Trust_In_Team_Members
init
Trust_In_Team_Members = 20

flow
Trust_In_Team_Members = +dt*IncInTrust_Of_Team


-dt*Dec_In_Trust

doc
Trust_In_Team_Members = The degree of trust members feel about each other.

Flow
BrnoutGeneration = ((Base_Burnout_Rt*Ef_PostLaunchHPM_On_Burn)*Limits_To_Burnout)*On_Off+1

Flow
Ch_in_Accept = Acceptance_Div_Ideas*Ef_MemTrust_On_AcceptDiv*EfComm_On_Acceptance*Fr_Diver

Flow
ChangeIn_Res = (AminTarget_Resources-Internal_Resources)/TimeToGet_Resources

doc
ChangeIn_Res = Standared stock adjustment process.

Flow
ChIn_CareForTeam_Members = (EffTrustOn_Care-Care_Respect)/TimeToCareForTeam_Membrs

Flow
Dec_In_Trust = (Trust_In_Team_Members/TmTodec_Trust)*EffCare_On_Decr_Trust

Flow
Dec_InTm_Func = Level_of_Tm_Func/Time_to_Dec_Tm_Func*Scale_Limit

Flow
DecIn_ExtResrs_1 = External_Resources/Tm_To_Dissp_ExRes

Flow
DecInTLrdrshpSkills = (Memb_Skills/TimeToLose_TLshpSkills)*On_Off

Flow
Decr_In_Acc = (Mutual_Accountability/Time_To_Lose_Account)*On_Off

Flow
DecreaseIn_Team_Ldr = Trust_In_Team_Ldr/Time_toDecTrust_In_T_Ldr

Flow
DecrIn_Vision = Clarity_Of_Vision/(TmTp_Dissp_Vision*Eff_Man_Bound_On_Keeping_vision)

Flow
DissOf_Brnout = (Burnout/Tm_to_DEDBurn)*Limits_To_Burnout

Flow
Erosion_Of_Goal = (Indicated_Goals/TimeDecGoal)*On_Off

Flow
Gain_In_MembComm = CoumpFrCommun*EfZeal_on_M_Commun*EfAcceptDiv_On_Commun*EfSkills_TechnoON_Commun*EfMemTrust_On_Commun*Member_Communication

Flow
Gain_Of_Commit = (Commitment*CompFrac_for_Committ*EfAm_On_Commit*EfftrustOn_Commit*EfNQROn_Commit*Ef_Kn_on_Commit*EfResrcOn_Commit)*ScaleLimit

Flow
GainZl = CompFracZl*Team_Zeal

Flow
Gen_Hire_Rate = (Desired_Team_Size-TeamSize)/Time_To_Fill_General_Positions

Flow
In_In_Empowr = Base_EmpRt*EfTot_Res_On_Emp*EfNewQualOn_Emp*Effect_MemSkls_On_Empr*Ef_InfClienta_and_Compet_on_Emp*Empowerment

Flow
Inc_Acc = (Base_Inc_In_Account*Ef_ClarityVis_On_Acc*EfCommitOnAccnt*Mutual_Accountability)*On_Off

Doc
Increase in Accountability

Flow
Inc_In_Team_Ldr = CompFrac_TrustLdr*EfControl_On_TrustTL*Ef_ResrceOnTrust*Trust_In_Team_Ldr

Flow
Inc_In_Tm_Funct = FracCompLTFunct*EfEffctv_Res_On_LTFunc*EfM_SkillsOn_LTFunc*EfInOnFunc*EfCommunOn_T_Func*EfClVis_On_LTFunc*Level_of_Tm_Func*Scale_Limit

Doc
Increase in Level of Team Functioning

Flow
IncIn__Inform = (Base_Inform_rate*Ef_Managing_BndryOn_Inf*Ef_LT_FunctOnInfCBase)*On_Off

Flow
IncIn_TSpirit = (CompFrac_IncT_Sprt*EfEmprw_OnTSpirit*EfCareOn_Spirt*Ef_LFunctOn_T_Spirit*Team_Spirit)*On_Off

Flow
IncInExt_Resrs = Ef_EmpOn_ExRes*Ef_LvTm_FunctOn_ExRes*EfInf_Clients_Compet_On_EXRes*EfLdrTrust_On_ExternRes*BaseInc_In_ExtRes*EfGain_OnExt_Res

Flow
IncInTrust_Of_Team = Trust_In_Team_Members*Ef_Acc_On_Trust_in_Tm_Membs*EfAccepDivtOn_Trust*EfKno_of_RoleOnMemTrust*EfGoal_On_Trust_Tm*BaseIncTrustInTeam

Flow
IncInVision = FractCompCVis*EMemb_Skills_On_Vision*EfKnowAbt_T_on_Vis*Clarity_Of_Vision

Flow
IncrIn_TLevAspir = (GoalMultiper*Indicated_Goals*EfMemTrust_On_Goals)*On_Off

Flow
IncrInTlrdsshpSkills = (BaseInc_InMembSkls*EfVis_On_MembSkills*EfManagBoundOn_Skls*GoalOnMemb_Skills*Memb_Skills)*On_Off

Flow
InIc_Perf = (Tm_Gap/(Time_For_Perf))*Ef_Rel_GainOnPerf*On_Off

aux
InR_Loyalt = CompFrac_Level*Loyalty*EfCareOn_Loyalty*Ef_Trust_InTmLdr_On_Loyalty*On_Off+.5

Flow
Learning_about_New_tasks = FrIncIn_learn*EfTaskRatioOnLearning*Knowledge*EfCommit_on_GainKn*Ef_Ambig_On_Know

Flow
Loss__of_Commit = (Commitment/CommitConstraint)*ScaleLimit

Flow
Loss_Of__Knowledge = (Knowledge/Time_To_Loose_Kn)*EfCommit_On_LossKn

Flow
Loss_Of_AcceptDiv = Acceptance_Div_Ideas/Constraint_On_Diversity

Flow
Loss_of_Commun = Member_Communication/Time_ToLoose_Commun

Flow
Loss_of_Mem_Amb = (Member_Ambiguity/Time_For_Loss_ofAmb)*EfKnOn_Am

Flow
Loss_of_Zl = Team_Zeal/(Constraint_onZeel*EfSpOnZeal1)

Flow
LossOf_Empowr = Empowerment/Tm_To_Lose_Emp_SlfEfficacy

Flow
LossOf_Inform = (InfAbout_CBaseYCompet/Tm_To_Lose_Inf)*On_Off

Flow
LossOf_Perf = (Performance/(Time_for_Loss_of_Perf))*EfAch_Opport_OnPerf*On_Off

Flow
LossOf_TSpirt = (Team_Spirit/Constraints_On_TSpirit)*On_Off

Flow
OutRt_Loyal = Loyalty/(ConstraintOn_Loyalty*Spirit_OnLoyalty)*On_Off

Flow
PreTeam_PerfChRate = (EfOldGapOnO_Perf/PerformTime)*EfTmSizeOn_PreTmPerf*(1-On_Off)

Flow
Quit_Fire_Rate = TeamSize/Av_Timeto_Leave

aux
Actual_HPM = On_Off*Indicated_HPM + (1-On_Off)*BaselIne_HrsPerMnth

aux
AltEf_Tm_Funct_On_ActlEffrrt = GRAPH(Level_of_Tm_Func,0,10,[1.23,1.23,1.18,1.07,0.99,0.87,0.74,0.67,0.64,0.64,0.64"Min:0;Max:1.5"])

doc
AltEf_Tm_Funct_On_ActlEffrrt = This is a bit different.  If the level of functioning is very low, it amplifies the effort it takes to get standard things done. The values of the table function would be above 1.0.   On the other hand, as the level of function goes up, then it decreases the number of hours it takes to do most things.

aux
AltHours_Mult = GRAPH(Tm_Gap,-6,4,[1,1,1,1.03,1.11,1.19,1.2,1.22,1.22,1.22,1.22"Min:0;Max:2"])

aux
Ambig_Gap = Tolorance_4_Ambig-Member_Ambiguity

aux
Av_Timeto_Leave = EF_LoyaltyOn_QuitRt

aux
CommitConstraint = GRAPH(Commitment,0,10,[200,199,178,64,34,18.5,12.3,7.3,4,3.1,1"Min:0;Max:200"])

aux
Constraint_On_Diversity = GRAPH(Acceptance_Div_Ideas,0,10,[39.4,38.9,38.1,37.2,34.2,28.5,22.3,7.2,3.6,2.5,1.5"Min:0;Max:40"])

aux
Constraint_onZeel =      GRAPH(Team_Zeal,0,10,[79.2,77,73.2,63,51.7,14.7,7.9,5.3,3.4,1.5,1.1"Min:0;Max:80"])

aux
ConstraintOn_Loyalty = GRAPH(Loyalty,0,10,[79.2,77,73.2,63,51.7,14.7,7.9,5.3,3.4,0.8,0.3"Min:0;Max:80"])

aux
ConstraintOnGoals = GRAPH(Indicated_Goals,0,10,[4000,4000,4000,4000,4000,4000,3980,3000,200,1,0.5"Min:0;Max:4000"])

aux
Constraints_On_TSpirit = GRAPH(Team_Spirit,0,10,[60,59.7,56.3,52.6,41.6,26.5,9.9,4.5,2.5,1.5,1"Min:0;Max:60"])

aux
Ef_Acc_On_Trust_in_Tm_Membs = GRAPH(Mutual_Accountability,0,10,[0.3,0.4,0.52,0.62,0.94,1.22,1.4,1.7,1.83,1.88,1.9"Min:0;Max:2"])

doc
Ef_Acc_On_Trust_in_Tm_Membs = You need Mutual Accountabilty to trust members. It ranges from less than 1.0 to a value greater than 1.0.

aux
Ef_Ambig_On_Know = GRAPH(Ambig_Gap,-100,10,[2.84,2.7,2.46,2.08,1.74,1.49,1.34,1.2,1.09,1.03,1"Min:0;Max:3"])

doc
Ef_Ambig_On_Know = The X axis deals with the Ambiguity Gap which is large at first and then goes down as the gap goes to zero.  It starts with a negative gap, because the  gap is defined as tolarance for Ambiguity - Ambiguity.  Since Member Role Ambiguity is large initially, the gap will almost always be negative for most runs.  The table function is large at first with a large negative gap, but goes down to 1.0 as the gap narrows.

aux
Ef_Burn_On_IGoal = GRAPH(Burnout,0,10,[1,1,1,1,0.99,0.89,0.8,0.73,0.65,0.61,0.61"Min:0;Max:1"])

doc
Ef_Burn_On_IGoal = The effect of Burnout does not come into play until Burnout gets to be around 40 units and then it decreases the Indicated Goal by making the table function take on values which are below 1.0.

aux
Ef_ClarityVis_On_Acc = GRAPH(Clarity_Of_Vision,0,10,[0.7,0.77,1,1.23,1.37,1.46,1.5,1.51,1.51,1.53,1.53"Min:0;Max:2"])

doc
Ef_ClarityVis_On_Acc = Low Clarity of vision, inhibits Accountability. The table function ranges from below 1.0 to > 1.0.

aux
Ef_EmpOn_ExRes = GRAPH(Empowerment,0,10,[0,0.02,0.83,0.93,0.96,0.97,0.98,0.99,0.99,1,1"Min:0;Max:1"])

doc
Ef_EmpOn_ExRes = This table function was like the table function associated with the effect of Empowerment on raising the goal.  With perfect empowerment, the value of the table function equal 1. while, when Empowerment is 0.0, the value of the table function equals 0.0, because, if you are not empowered, you will not go after external resources.

aux
Ef_InfClienta_and_Compet_on_Emp = GRAPH(InfAbout_CBaseYCompet,0,10,[1,1,1,1.01,1.05,1.14,1.17,1.17,1.17,1.17,1.17"Min:0;Max:1.25"])

doc
Ef_InfClienta_and_Compet_on_Emp = At first, information does not have much to do with empowerment, but as information about competition and outside resources increases, it can increase the feeling of empowerment.

aux
Ef_Kn_on_Commit = GRAPH(Knowledge,0,10,[1,1,1,1.02,1.08,1.16,1.24,1.3,1.39,1.41,1.41"Min:0;Max:2"])

doc
Ef_Kn_on_Commit = If you have no knowledge about the team concept, you can still increase your Commitment if other factors come into play. So here the multiplier is 1.0 at low decrees of Knowledge and above 1 for more extreme degrees of Knowledge.

aux
Ef_LFunctOn_T_Spirit = GRAPH(Level_of_Tm_Func,0,10,[0.65,0.65,0.73,0.77,0.9,1.1,1.85,2.25,2.4,2.45,2.45"Min:0;Max:4"])

doc
Ef_LFunctOn_T_Spirit = Low levels of Functioning inhibit spirit, while high levels generate spirit.  Table function starts below 1.0 and then at high levels of functioning goes above 1.0.

aux
EF_LoyaltyOn_QuitRt = GRAPH(Loyalty,0,10,[13,16,19,26,71,204,464,555,586,600,600"Min:0;Max:600"])

doc
EF_LoyaltyOn_QuitRt = Table function starts with low average time and increases time to loose team members as loyalty gets high.

aux
Ef_LT_FunctOnInfCBase = GRAPH(Level_of_Tm_Func,0,10,[1,1,1,1.02,1.08,1.22,1.37,1.47,1.49,1.49,1.49"Min:0;Max:1.74"])

doc
Ef_LT_FunctOnInfCBase = At first, there is no effect of team functioning, but then as it increases, it helps to get information to the team.

aux
Ef_LvTm_FunctOn_ExRes = GRAPH(Level_of_Tm_Func,0,10,[1,1,1.06,1.2,1.43,2.94,3.25,3.3,3.3,3.3,3.3"Min:0;Max:4"])

doc
Ef_LvTm_FunctOn_ExRes = If the team has a low Level Of Functioning, then they certainly will not increase in getting external forces.  I let that value be equal to 1.0, assuming that the External Resource level variable was represented by a first order material delay in equilibrium.  As the level of functioning goes up, the value of the table function goes up beyond 1.0.

aux
Ef_Managing_BndryOn_Inf = GRAPH(Man_Bound,0,10,[1,2.2,4.2,7,10.1,12.2,12.9,13,13,13,13"Min:0;Max:13"])

doc
Ef_Managing_BndryOn_Inf = Low degrees of managing the boundary do not affect the baseline, but as Managing the Boundary increases, this type of guidance is quite helpful in getting information to the team.

aux
Ef_MemTrust_On_AcceptDiv = GRAPH(Trust_In_Team_Members,0,10,[0,0.08,0.2,0.61,0.9,1.03,1.25,1.64,1.82,1.87,1.89"Min:0;Max:2"])

doc
Ef_MemTrust_On_AcceptDiv = No trust implies implies no growth of Acceptance of Diverse Ideas.  Goes from 0.0 to a value greater than 1.0.

aux
Ef_PostLaunchHPM_On_Burn = GRAPH(Indicated_HPM,100,30,[1,1,1,1.58,1.83,1.979,2,2,2,2,2"Min:0;Max:2"])

doc
Ef_PostLaunchHPM_On_Burn = At first, low number of hours do not cntribute to burnout, but as hours increase, burnout will accumlate.  The table function starts at a value of 1.0 initilaly, but then goes up to a value of 2 as the number of work hours increases.

aux
Ef_Rel_GainOnPerf = GRAPH(Rel_GainHrs,-0.5,0.1,[0.67,0.7,0.71,0.76,0.84,1,1.1,1.13,1.18,1.22,1.23"Min:0;Max:1.5"])

doc
Ef_Rel_GainOnPerf = This is one of the variables which assumes that if you do not have gain hours per month,  you cannot increase beyond what you would have with the standard effort.  When the table function goes between 0 and 1, the factor inhibits the growth of the state variable. When it goes above, 1.0 it helps to increase the input rate.

aux
Ef_ResrceOnTrust = GRAPH(Resource_Ratio,0,0.1,[0,0.02,0.03,0.03,0.04,0.05,0.14,0.92,1,1,1,1,1"Min:0;Max:1"])

doc
Ef_ResrceOnTrust = This is the main source of variation in generating trust in the team leader.  Will there be resources available to get started.  When the resource ratio, equals 0.0, the multiplier = 0.0.  The range goes from 0.0 to 1.0.

aux
Ef_TmFuncOn_IndHPM = GRAPH(Level_of_Tm_Func,0,10,[1,1,1,1,1,0.86,0.83,0.82,0.81,0.79,0.79"Min:0;Max:1.5"])

doc
Ef_TmFuncOn_IndHPM = In this alternative, high level of functioning helps to make the team more efficient and thus free up more hours for  exploring external funding sources.

aux
Ef_Trust_InTmLdr_On_Loyalty = GRAPH(Trust_In_Team_Ldr,0,10,[0.75,0.77,0.83,0.91,1.15,1.3,1.42,1.52,1.57,1.63,1.63"Min:0;Max:2"])

doc
Ef_Trust_InTmLdr_On_Loyalty = Trust is a substitute for liking the leader.  The table function goes from below 1.0 to being above 1.0.

aux
EfAcc_On_Goals = GRAPH(Mutual_Accountability,0,10,[-0.01,-0.0082,-0.0062,-0.0044,-0.0025,-0.0001,0.0022,0.0048,0.007,0.0088,0.01"Min:-0.01;Max:0.01"])

doc
EfAcc_On_Goals = Here we have a sense of mutual accountabilty.  Low values of accountability leads to lowering the goal.  High values of accountability, on the othter hand, raises the goal or the bar.

aux
EfAccepDivtOn_Trust = GRAPH(Acceptance_Div_Ideas,0,10,[0.76,0.79,0.83,1.02,1.25,1.54,1.62,1.64,1.64,1.64,1.64"Min:0;Max:2"])

aux
EfAcceptDiv_On_Commun = GRAPH(Acceptance_Div_Ideas,0,10,[0.68,0.68,0.69,0.71,0.76,0.83,0.99,1.04,1.19,1.21,1.27"Min:0;Max:1.5"])

doc
EfAcceptDiv_On_Commun = Low acceptance of diverse ideas tends to inhibit communication while high acceptance tends to increase communication.

aux
EfAch_Opport_OnPerf = GRAPH(Man_Bound,0,10,[1,1,1,1,0.99,0.85,0.58,0.34,0.07,0,0"Min:0;Max:1"])

aux
EfAm_On_Commit = GRAPH(Member_Ambiguity,0,0.1,[1,1,1,0.99,0.86,0.83,0.83,0.83,0.83,0.83,0.83"Min:0;Max:1"])

doc
EfAm_On_Commit = high degrees of Ambiguity inhibit the growth of Commitment, but low degrees do not have much if any effect on increasing Commitment.

aux
EfCareOn_Loyalty = GRAPH(Care_Respect,0,10,[0.24,0.27,0.33,0.62,1.01,1.13,1.21,1.4,1.83,1.88,1.89"Min:0;Max:2"])

doc
EfCareOn_Loyalty = Care for team members is a very proccess for loyalty, but zero care can still generate some loyaty. Table function starts below 1.0 and extends above 1.0.

aux
EfCareOn_Spirt = GRAPH(Care_Respect,0,10,[0.29,0.36,0.41,0.49,0.71,1,1.07,1.25,2.26,2.39,2.43"Min:0;Max:3"])

doc
EfCareOn_Spirt = If you do not care for each other, then there cannot be mcuh growth in team spirit. The range of the table function goes from below 1.0 to above 1.0.

aux
EfClVis_On_LTFunc = GRAPH(Clarity_Of_Vision,0,10,[0.41,0.79,0.92,0.96,0.98,1,1.33,1.97,2.18,2.26,2.29"Min:0;Max:3"])

doc
EfClVis_On_LTFunc = The Clarity of Vision inhibits the growth of Level of T Funcion when it is low, but enhances the growth when Clarity of Vision is high.  The table function ranges from  below 1.0 to above 1.0.

aux
EfComm_On_Acceptance = GRAPH(Member_Communication,0,10,[0.72,0.72,0.74,0.76,0.81,0.92,1.2,1.29,1.35,1.38,1.39"Min:0;Max:1.5"])

doc
EfComm_On_Acceptance = Goes from a pt below 1.0 to a pt above 1.0

aux
EfCommit_on_GainKn = GRAPH(Commitment,0,10,[0,0.01,0.04,0.06,0.15,0.71,1.2,1.41,1.57,1.61,1.62"Min:0;Max:2"])

doc
EfCommit_on_GainKn = This is one of the variables which assumes that if you do not have Team Commitment, you cannot learn anything.  When the table function goes between 0 and 1, the factor inhibits the growth of the state variable. When it goes above, 1.0 it helps to increase the input rate.

aux
EfCommit_On_LossKn = GRAPH(Commitment,0,10,[1.27,1.25,1.17,1,1,1,1,1,1,1,1"Min:0;Max:1.5"])

doc
EfCommit_On_LossKn = I have assumed that low commitment, helps to increase the loss of Knowledge.

aux
EfCommitOnAccnt = GRAPH(Commitment,0,10,[0,0.12,0.25,0.63,1.13,1.56,1.81,1.85,1.9,1.93,1.94"Min:0;Max:2"])

doc
EfCommitOnAccnt = Commitment ot the team is a necessary condition.  No Commitment implies no increase in Mutual Accountability.  The multiplier ranges from 0.0 to a number above 1.0

aux
EfCommunOn_T_Func = GRAPH(Member_Communication,0,10,[0.79,0.78,0.79,0.81,0.88,1,1.4,2.19,2.33,2.39,2.42"Min:0;Max:3"])

doc
EfCommunOn_T_Func = Low degrees of Communication inhibit T Functioning, while high degrees of Communiton enhances it .  The table function starts below 1.0 and then rises above 1.0 for high degrees of Communiication.

aux
EfControl_On_TrustTL = GRAPH(Memory_Of_Control,0,10,[1,1,1,0.94,0.68,0.35,0.18,0.1,0.1,0.1,0.1"Min:0;Max:1"])

doc
EfControl_On_TrustTL = If control is very high, then the members are getting a direct and symbolic message that this will not be a self-directed group.  How can they trust the leader when he or she is still controlling?  Thus, if control is low, the multiplier =1.0, but goes down as control increases toward 100.

aux
EfEffctv_Res_On_LTFunc = GRAPH(Effective_Tot_Res,0,10,[0,0.03,0.03,0.06,0.1,0.17,0.29,0.46,0.74,1.25,1.32"Min:0;Max:2"])

doc
EfEffctv_Res_On_LTFunc = No resources makes the table function equal to 0.0.  High or adequate resources makes the table function attain a value above 1.0.

aux
EfEfftv_Tot_Rs_OnAspir = GRAPH(Effective_Tot_Res,0,10,[0.76,0.83,0.85,0.87,0.9,0.94,0.96,0.98,1.08,1.15,1.17"Min:0;Max:1.5"])

doc
EfEfftv_Tot_Rs_OnAspir = If one does not have the resources to do the job, the Level of Aspiration would be lower than the intended goal.  The table function value would be less than 1.0.  On the other hand, if the resources are adequate, then the Level of as piration will be above the intended goal.

aux
EfEmp_On_Goals = GRAPH(Empowerment,0,10,[0,0.02,0.03,0.03,0.04,0.07,0.17,0.85,0.95,0.99,1"Min:0;Max:1"])

doc
EfEmp_On_Goals = Empowerment allows one to do something. It is self-efficacy.  On the other hand, if empowerment is low, then the multiplier will be below 1.0.  We have set the max value at 1.0, at perfect empowerment.  If you are empowered, you still may not increase the goal, so we set the maximimum value at 1.0,instead of being greater than 1.0.

aux
EfEmprw_OnTSpirit = GRAPH(Empowerment,0,10,[0.77,0.78,0.81,0.87,1,1.3,2.15,2.36,2.45,2.45,2.45"Min:0;Max:4"])

aux
Eff_Man_Bound_On_Keeping_vision = GRAPH(Man_Bound,0,10,[0.61,0.62,0.67,0.83,1,1.05,1.08,1.08,1.1,1.11,1.13"Min:0;Max:2"])

doc
Eff_Man_Bound_On_Keeping_vision = Managing the Boundaries helps to keep and guide Clarity of Vision.  It ranges from below 1.0 to above 1.0.  When the table function is less than 1.0 it decrease Vision, while it will increase or maintain Clarity of Vision for high values of Managing the Boundary.

aux
EffCare_On_Decr_Trust = GRAPH(Care_Respect,0,10,[1,1,1,0.99,0.97,0.96,0.96,0.95,0.94,0.94,0.94"Min:0;Max:1"])

doc
EffCare_On_Decr_Trust = This is a discounting processes.  When care is high, the multiplier is less than 1.0. It  slows down the loss of trust.

aux
Effect_MemSkls_On_Empr = GRAPH(Memb_Skills,0,10,[0.28,0.4,0.99,1.2,1.51,1.72,1.89,1.98,2.02,2.08,2.08"Min:0;Max:2.5"])

doc
Effect_MemSkls_On_Empr = One can become empowered slightly with no Skills, but I the value of the multiplier is close to 0.0.  It starts below 1.0 and eventually goes above 1.0.  Thus it inhibits and enhances growth.

aux
Effective_Tot_Res = TrustFilterOn_Tot_Res*Total_Resources

aux
EffTrustOn_Care = GRAPH(Trust_In_Team_Members,0,10,[4,7,9,15,24,35,47,59,80,90,100"Min:0;Max:100"])

aux
EfftrustOn_Commit = GRAPH(Trust_In_Team_Ldr,0,10,[0.04,0.14,0.48,0.95,1.12,1.19,1.24,1.25,1.3,1.31,1.31"Min:0;Max:2"])

doc
EfftrustOn_Commit = This is one of the variables which assumes that if you do not have trust in the team leader, you cannot become Committed.  When the table function goes between 0 and 1, the factor inhibits the growth of the state variable. When it goes above, 1.0 it helps to increase the input rate.

aux
EfGain_OnExt_Res = GRAPH(Rel_GainHrs,-0.5,0.1,[0,0,0,0,0,0,1.57,1.76,1.85,1.87,1.87"Min:0;Max:2"])

doc
EfGain_OnExt_Res = Relative gain represent slack hours to apply to gaining additional resources that are external to the company.

aux
EfGoal_On_Trust_Tm = GRAPH(New_QualityRatio,0.6,0.1,[0.62,0.65,0.7,1.04,1.5"Min:0;Max:1.5"])

doc
EfGoal_On_Trust_Tm = In this situation, reaching the goal has a positive effect on generating in fellows team members.  In this case, a low ratio tends to inhibit trust, but as the team moves toward reaching the goal,  trust increases.

aux
EfInf_Clients_Compet_On_EXRes = GRAPH(InfAbout_CBaseYCompet,0,10,[0.26,0.31,0.36,0.41,0.53,0.9,1,1,1,1,1"Min:0;Max:1"])

doc
EfInf_Clients_Compet_On_EXRes = Low information can inhibit the generation of External Resources.  High information can enhanse the generation of External Resources.

aux
EfInOnFunc = GRAPH(InfAbout_CBaseYCompet,0,10,[1,1,1.01,1.07,1.3,1.7,1.84,1.95,1.98,2,2"Min:0;Max:2"])

doc
EfInOnFunc = The table fuction starts at 1.0 and then goes above a value of one for high degrees of Extennal information.

aux
EfKno_of_RoleOnMemTrust = GRAPH(Knowledge,0,10,[0.5,0.61,0.7,0.84,0.99,1.21,2.11,2.46,2.6,2.67,2.69,2.69"Min:0;Max:3"])

doc
EfKno_of_RoleOnMemTrust = Ranges from below 1.0 to above 1.0

aux
EfKnOn_Am = GRAPH(Knowledge,0,10,[0,0.05,0.13,0.25,0.5,1,2.4,4.3,17.8,19.5,20"Min:0;Max:20"])

aux
EfKnowAbt_T_on_Vis = GRAPH(Knowledge,0,10,[1,1,1.08,1.15,1.27,1.56,1.85,2.05,2.15,2.17,2.17"Min:0;Max:3"])

doc
EfKnowAbt_T_on_Vis = Lack of Knowldge does not have an effect on clarity of vision, but as Knowledfe increases, it has more impact on Clarity of vision.  The table function ranges from 1.0 to above 1.0.

aux
EfLdrTrust_On_ExternRes = GRAPH(Trust_In_Team_Ldr,0,10,[0,0.02,0.04,0.08,0.14,0.21,0.32,0.73,0.88,0.94,1"Min:0;Max:1"])

doc
EfLdrTrust_On_ExternRes = No trust in leader generates 0.0 table function value. As trust in leader increases, then the value of the table function approaches 1.0.

aux
EfM_SkillsOn_LTFunc = GRAPH(Memb_Skills,0,10,[1,1,1,1.01,1.04,1.08,1.23,1.4,1.49,1.51,1.52"Min:0;Max:2"])

doc
EfM_SkillsOn_LTFunc = The table fuction starts at 1.0 and then goes above a value of one for high degrees of Skills.

aux
EfManagBoundOn_Skls = GRAPH(Man_Bound,0,10,[0.39,0.42,0.45,0.6,1.01,1.1,1.34,1.45,1.49,1.49,1.49"Min:0;Max:1.5"])

doc
EfManagBoundOn_Skls = Poor management can waste time getting  the wrong skills.  This falls under both guidence and opportunity.  The range of the table is below 1.0 to greater than 1.0.

aux
EfMemTrust_On_Commun = GRAPH(Trust_In_Team_Members,0,10,[0.63,0.65,0.71,0.95,1.03,1.13,1.35,1.65,1.84,1.9,1.91"Min:0;Max:2"])

doc
EfMemTrust_On_Commun = Lack of trust will inhibit communication, while high degrees of  T member trust will enhance Communicaton.

aux
EfMemTrust_On_Goals = GRAPH(Trust_In_Team_Members,0,10,[0,0,0.03,0.06,0.63,0.95,1,1,1,1,1"Min:0;Max:1"])

doc
EfMemTrust_On_Goals = This is one of the variables which assumes that if you do not have trust in the team, you cannot raise the goal.  When the table function goes between 0 and 1, the factor inhibits the growth of the state variable. When it goes above 1.0, it helps to increase the input rate.

aux
EfNewQualOn_Emp = GRAPH(New_QualityRatio,0.6,0.1,[1,1.15,2.08,2.21,2.21"Min:0;Max:2.5"])

doc
EfNewQualOn_Emp = Empowerment increases with success.  Thus, when the ratio = 0.0, the multiplier = 1.0, and as it increases to 1.0, success, it increases empowement.  This is the opposite of the situation where the small ratios generate action to close the gap.

aux
EfNQROn_Commit = GRAPH(New_QualityRatio,0.6,0.1,[8,7.74,7.28,1.89,1.27,1.11,1.08,1.05,1.04,1.02,1.01"Min:0;Max:8"])

doc
EfNQROn_Commit = A low ratio implies that the organization is in big touble.  It needs to do something to "bridge the gap", so to speak.  In this situation, a low ratio drives the increase in Commitment.  The smaller the ratio , the more likely the team will be commited to the team concept.  When the New_Quality Ratio gets to be 1.0, then there is still some influence of Performance, but this is only  a little bit above 1.0.

aux
EfOldGapOnO_Perf = GRAPH(Perf_Gap,-50,10,[0,0,0,0,0,0,10,20,30,40,50"Min:0;Max:50"])

doc
EfOldGapOnO_Perf = This table function is set up so that if Performance is below the Old Standard, the system will try to raise performance.  On the other hand, for negative values of the Perf_Gap, the system will not try to decrease performance down to the old standard.  There will not be an effort to ever lower performance.

aux
EfResrcOn_Commit = GRAPH(Effective_Tot_Res,0,10,[0,0.03,0.04,0.1,0.16,0.34,0.54,2.18,2.24,2.26,2.25"Min:0;Max:3"])

doc
EfResrcOn_Commit = This is one of the variables which assumes that if you do not have Resources, you cannot become Committed.  When the table function goes between 0 and 1, the factor inhibits the growth of the state variable. When it goes above, 1.0 it helps to increase the input rate.

aux
EfSkills_TechnoON_Commun = GRAPH(Memb_Skills,0,10,[1,1.01,1.1,1.18,1.32,1.42,1.48,1.5,1.5,1.5,1.5"Min:0;Max:1.5"])

doc
EfSkills_TechnoON_Commun = The table function starts at 1.0 and increases as Sills increase.

aux
EfSp_OnZeal = GRAPH(Team_Spirit,0,10,[0,0.1,0.2,0.4,0.8,1.8,8,10.8,11.7,12.3,12.4"Min:0;Max:15"])

doc
EfSp_OnZeal = No Spirit, no Zeal.  Low values of spirit inhibit Zeal, but large values of Spirit enhance Zeal.

aux
EfSpOnZeal1 = GRAPH(Team_Spirit,0,10,[0,0.1,0.2,0.4,0.8,1.8,8,10.8,11.7,12.3,12.4"Min:0;Max:13"])

aux
EfTaskRatioOnLearning = GRAPH(Task_Ratio,0,0.1,[1,1,1,1,1,1,1,1,0.32,0.11,0"Min:0;Max:1"])

doc
EfTaskRatioOnLearning = This is one way to reach a stable point with a positive loop.  As the task ratio goes near 1.0, namely when all the tasks are learned, the table function goes to 10.0.  This prevents it from going negative.

aux
EfTm_Size_On_HrperMnt = GRAPH(Rel_Tm_Size,0,0.1,[1.27,1.25,1.24,1.22,1.19,1.11,1.06,1.04,1.03,1.01,1,1,1"Min:0;Max:1.4"])

doc
EfTm_Size_On_HrperMnt = When the relative size of the team is low, then the number of hours goes up.  As the relative size of the team goes down to one, the number of hours goes down to 1.0.  Finally, when the relative size is ideal, then the multiplier equal 1.0.

aux
EfTmSizeOn_PreTmPerf = GRAPH(Rel_Tm_Size,0,0.1,[0.54,0.56,0.57,0.59,0.62,0.65,0.67,0.74,0.82,0.91,1,1.09,1.1"Min:0;Max:1.5"])

aux
EfTot_Res_On_Emp = GRAPH(Effective_Tot_Res,0,10,[0,0.04,0.1,0.19,0.28,0.37,0.54,0.75,0.97,2.36,2.46"Min:0;Max:2.5"])

doc
EfTot_Res_On_Emp = No resources, shuts off the input to the Empowerment variable. Resources below 1.0 inhibit empowerment and then after  resources get to be about 80, the  multiplier becomes greater than 1.0

aux
EfVis_On_MembSkills = GRAPH(Clarity_Of_Vision,0,10,[0.59,0.62,0.75,1.02,1.02,1.07,1.11,1.29,1.8,1.96,2"Min:0;Max:2"])

doc
EfVis_On_MembSkills = This takes the function of guidence.  The table function ranges from below 1.0 to above 1.0.

aux
EfZeal_On_Brn = GRAPH(Team_Zeal,0,10,[1,0.97,0.94,0.85,0.76,0.43,0.23,0.16,0.09,0.04,0.04"Min:0;Max:1"])

doc
EfZeal_On_Brn = Low values of Zeal do not have an effect on the loss of burnout.  However, as Zeal increases, the value of the table function decreases below 1.0, cutting down the product of the time constant and the table function.  A decrease in this product, quickens the loss of Burnout.

aux
EfZeal_on_M_Commun = GRAPH(Team_Zeal,0,10,[1,1,1.01,1.04,1.08,1.16,1.25,1.32,1.35,1.36,1.37"Min:0;Max:1.5"])

doc
EfZeal_on_M_Commun = Zeel has no effect at first, but later helps to increase Communication. The table function starts at 1.0 and then goes above 1.0.

aux
EMemb_Skills_On_Vision = GRAPH(Memb_Skills,0,10,[0.98,1,1,1.03,1.23,1.61,2,2.67,2.94,2.99,3"Min:0;Max:3"])

doc
EMemb_Skills_On_Vision = Lack of skills does not have an effect on clarity of vision, but as Skills increase, it has more impact on Clarity of vision.  The table function ranges from 1.0 to above 1.0.

aux
GoalMultiper = EfAcc_On_Goals*EfEmp_On_Goals

doc
GoalMultiper = Note the product of these two gives one the abiltiy to show that if empowerment is low and a sense of accountability is low, then the bar is not raised,  And, if empowerment is high and accountabilyty is low, then the bar is also not raised.

aux
GoalOnMemb_Skills = GRAPH(New_QualityRatio,0.6,0.1,[1.99,1.84,1.63,1.35,1.17,1.06,1.02,1.02"Min:0;Max:2"])

doc
GoalOnMemb_Skills = New QualityRatio motivates getting the skills necessary to do the job well.  Low values of the ratio stimulate getting skills.

aux
HoursMult = GRAPH(Tm_Gap,-6,4,[1,1,1,1.03,1.06,1.07,1.07,1.07,1.07,1.07,1.07"Min:0;Max:2"])

aux
Hrs_Gained = BaselIne_HrsPerMnth-Actual_HPM

aux
Indicated_HPM = BaselIne_HrsPerMnth*EfTm_Size_On_HrperMnt*AltHours_Mult*Ef_TmFuncOn_IndHPM

aux
Limits_To_Burnout = GRAPH(Burnout,0,10,[1,1,1,1,1,1,1,1,1,0.9,0"Min:0;Max:1"])

doc
Limits_To_Burnout = A way of keeping Burnout from going over 100 untis.

aux
Lv_AspirationPost_Launch = Indicated_Goals*EfEfftv_Tot_Rs_OnAspir

aux
Man_Bound = GRAPH(TimeSeries,0,10,[5,5,13,22,40,83,89,91,92,92,92"Min:0;Max:100"])

doc
Man_Bound = First one should note that in this team dynamics model, the variable is exogenous. It is an internal variable in the leadership model, i.e., in the leadership model its values over time are determined internally. Man_Bound (Managing the boundaries) is a variable that represents the team leader's ability to work well with other parts of the organization, help the team get information from clients and customers, as well as information about relevant competitor behavior.

aux
Memory_Of_Control = GRAPH(TimeSeries_1,0,10,[75,53.56,39.9,28.4,20.86,15.26,11.17,8.19,5.99,4,3"Min:0;Max:100"])

aux
Net_Gain_Burnout = BrnoutGeneration-DissOf_Brnout

aux
New_QualityRatio = Performance/(Lv_AspirationPost_Launch +.001)

doc
New_QualityRatio = This is a relative performance index under realistically changed competitive conditions.

aux
OldQualityRatio = Performance/(Old_Standard + .001)

doc
OldQualityRatio = Relative performance under conditions insulated from more general competition.  

aux
On_Off = 0 +STEP(1,24)

doc
On_Off = A necessary external switch to deal with the pre-formation and formation stages of the evolution of the team.  Note that this team model can be connected with the team leadership model that internally determines if and when actual teams come on line.

aux
Perf_Gap = Old_Standard - Performance

aux
Performance_Goal = On_Off*Lv_AspirationPost_Launch + (1-On_Off)*Old_Standard

aux
PerformTime = GRAPH(Memory_Of_Control,0,10,[24,24,23,22.9,21.8,20.4,17.5,11.9,5.1,2,1.7"Min:0;Max:60"])

doc
PerformTime = If the performance is below the old standard, the teams perception of the supervisor's control takes over, and the team is "whipped" into trying to go back to the old standard.  Thus when perceived control is high, the delay is small.  On the other hand, under the old standard, when the perceived control is low, the lag is longer.  

aux
Rel_GainHrs = Hrs_Gained/BaselIne_HrsPerMnth

doc
Rel_GainHrs = Essential the percentage gained or lossed over or under the baseline hours per month which usually will be set at 160 hours per month per person.

aux
Rel_Tm_Size = TeamSize/Desired_Team_Size

aux
Resource_Ratio = Total_Resources/Initial_Team_Expected_Res

aux
Scale_Limit = GRAPH(Level_of_Tm_Func,0,10,[1,1,1,1,1,1,1,1,0.97,0.93,0"Min:0;Max:1"])

aux
ScaleLimit = GRAPH(Commitment,0,10,[1,1,1,1,1,1,1,1,1,0.93,0"Min:0;Max:1"])

doc
ScaleLimit = Commitment is explosive!

aux
Spirit_OnLoyalty = GRAPH(Team_Spirit,0,10,[1,1,1.02,1.02,1.05,1.33,1.58,2.58,5.18,5.94,6"Min:0;Max:7"])

doc
Spirit_OnLoyalty = Low degree of Spirit has no effect on loyalty, but it tends to shut off decreases in loyalty as Spirit increases.  The table function starts at 1.0 and increases above that value as Spirit increases.

aux
Task_Ratio = Knowledge/Total_Roles

aux
Time_to_Dec_Tm_Func = GRAPH(Level_of_Tm_Func,0,10,[41.8,41,38.4,36.1,32.5,26.5,15.3,10.4,5.5,3.8,1.8"Min:0;Max:45"])

doc
Time_to_Dec_Tm_Func = If the system were to stop the team from functioning, then the level of functioning would decline with an average time of 9 months, assuming that the team were to go on for that time

aux
Time_To_Lose_Account = GRAPH(Mutual_Accountability,0,10,[23.9,25,24.8,24.6,21.5,18.5,15.2,10.4,3.8,1.5,1.2"Min:0;Max:25"])

aux
Time_toDecTrust_In_T_Ldr = GRAPH(Trust_In_Team_Ldr,0,10,[24.4,24.2,23.9,22.1,19.1,14,7.2,4.6,3.4,2.9,2.5"Min:0;Max:25"])

aux
Time_ToLoose_Commun = GRAPH(Member_Communication,0,10,[44.8,44.8,44.4,42.2,35.4,23.6,12.3,9.1,4.7,3.2,2.25"Min:0;Max:45"])

aux
TimeDecGoal = Ef_Burn_On_IGoal*ConstraintOnGoals

aux
TimeSeries = TIME

aux
TimeSeries_1 = TIME

aux
TimeToLose_TLshpSkills = GRAPH(Memb_Skills,0,10,[23.9,23.8,23.7,22.6,21.5,18.5,14,10.4,4.5,2.4,1.7"Min:0;Max:25"])

aux
Tm_Gap = Performance_Goal-Performance

aux
Tm_to_DEDBurn = BsTime_ToDec_Burn*EfZeal_On_Brn

aux
Tm_To_Lose_Emp_SlfEfficacy = GRAPH(Empowerment,0,10,[23.9,23.8,23.7,22.6,21.5,18.5,14,6,1.5,0.9,0.4"Min:0;Max:40"])

aux
TmTodec_Trust = GRAPH(Trust_In_Team_Members,0,10,[23.8,23.8,23.7,23.5,23.5,22.6,21.3,18.5,2.2,1.2,0.9"Min:0;Max:25"])

aux
TmTp_Dissp_Vision = GRAPH(Clarity_Of_Vision,0,10,[23.9,23.8,23.7,22.6,21.5,18.5,14,8.5,5.1,3.2,1.9"Min:0;Max:25"])

aux
Total_Resources = External_Resources+Internal_Resources

aux
TrustFilterOn_Tot_Res = GRAPH(Trust_In_Team_Ldr,0,10,[0,0.11,0.42,0.69,0.8,0.9,0.99,1,1,1,1"Min:0;Max:1"])

doc
TrustFilterOn_Tot_Res = Generates the validity of the resources filtered by the trust the members have on resources coming down the line.  No trust, filters out what comes down the pipeline.

const
AminTarget_Resources = 85

doc
AminTarget_Resources = This figure may or not may be  based on knowledge of the amount of resources needed to make the team work by upper management.

const
Base_Burnout_Rt = 4

const
Base_EmpRt = .09

const
Base_Inc_In_Account = .06

const
Base_Inform_rate = 1.25

const
BaseInc_In_ExtRes = .415

const
BaseInc_InMembSkls = .15

const
BaseIncTrustInTeam = .08

const
BaselIne_HrsPerMnth = 160

const
BsTime_ToDec_Burn = 16

const
CompFrac_for_Committ = .97

const
CompFrac_IncT_Sprt = .06

const
CompFrac_Level = .15

const
CompFrac_TrustLdr = .4

const
CompFracZl = .07

const
CoumpFrCommun = .08

const
Desired_Team_Size = 10

const
Fr_Diver = .23

const
FracCompLTFunct = .2

const
FractCompCVis = .06

const
FrIncIn_learn = .3

const
Initial_Team_Expected_Res = 85

doc
Initial_Team_Expected_Res = This figure corresponds to the amount of resources told to the supervisor gy upper management.

const
Old_Standard = 40

doc
Old_Standard = This is the old standard of performance that is not competitive in the general, perhaps global market.

const
Time_for_Loss_of_Perf = 100

const
Time_For_Loss_ofAmb = 100

const
Time_For_Perf = 6

const
Time_To_Fill_General_Positions = 6

const
Time_To_Loose_Kn = 35

const
TimeToCareForTeam_Membrs = 6

const
TimeToGet_Resources = 6

const
Tm_To_Dissp_ExRes = 6

const
Tm_To_Lose_Inf = 4

doc
Tm_To_Lose_Inf = The time constant probably should be fairly short if the team is somewhat cut off from getting information concerning the customer base.  

const
Tolorance_4_Ambig = 0

const
Total_Roles = 100

Run # 2.  Changes Due to Inadequate  resource run

init
Internal_Resources = 25

Run #3. Changes Due to Ineffective Team Leadership 

aux
Memory_Of_Control =    

GRAPH(TimeSeries_1,0,10,[80,78,73,72,72,71,71,71,71,71,71"Min:0;Max:100"])

const
Man_Bound = 3

doc
Man_Bound = First one should note that in this team dynamics model, the variable is exogenous. It is an internal variable in the leadership model, i.e., in the leadership model its values over time are determined internally. Man_Bound (Managing the boundaries) is a variable that represents the team leader's ability to work well with other parts of the organization, help the team get information from clients and customers, as well as information about relevant competitor behavior.

